
<program>               ::= <fbes> <inst> <rules> 
 
<fbes>                  ::= <fbe> 
                         |  <fbe> <fbes> 
 
<fbe>                   ::= FBE <id> <fbe_body> END_FBE 
                         |  FBE <fbe_body> END_FBE 
 
<id>                    ::= ID 
                         |  ID POINT ID 
 
<fbe_body>              ::= <decl_attributes> <decl_methods> 
 
<decl_attributes>       ::= ATTRIBUTES <atributes> END_ATTRIBUTES 
 
<atributes>             ::= <attributes_body> 
                         |  <attributes_body> <attributes> 
 
<attributes_body>       ::= <type> <id> <value> 
 
<type>                  ::= BOOLEAN 
                         |  INTEGER 
                         |  REAL 
                         |  STRING 
                         |  <id> 
 
<value>                 ::= NUMBER 
                         |  <boolean> 
                         |  <id> 
 
<boolean>               ::= TRUE 
                         |  FALSE 
 
<decl_methods>          ::= METHODS <methods> END_METHODS 
 
<methods>               ::= <method_body> 
                         |  <method_body> <methods> 
 
<method_body>           ::= METHOD <id> LP <id> ASSIGN <value> RP 
 
<inst>                  ::= INST <declarations> END_INST 
 
<declarations>          ::= <declaration> 
                         |  <declaration> <declarations> 
 
<declaration>           ::= <type> <ids> 
 
<ids>                   ::= <id> 
                         |  <id> COMMA <ids> 
 
<rules>                 ::= <rule> 
                         |  <rule> <rules> 
 
<rule>                  ::= RULE <rule_body> END_RULE 
                         |  RULE <id> <rule_body> END_RULE 
 
<rule_body>             ::= <decl_condition> <decl_action> 

<decl_condition>        ::= CONDITION <condition_body> END_CONDITION 
                         |  CONDITION <id> <condition_body> END_CONDITION 
 
<condition_body>        ::= <subcondition> <operator> <condition_body> 
                         |  <subcondition> 
 
<operator>              ::= AND 
                         |  OR 
 
<subcondition>          ::= SUBCONDITION <subcondition_body> 
                                  END_SUBCONDITION 
                         |  SUBCONDITION <id> <subcondition_body> 
                                  END_SUBCONDITION 
 
<subcondition_body>     ::= <premise> AND <subcondition_body> 
                         |  <premise> 
 
<premise>               ::= PREMISE <exp> 
                         |  PREMISE <id> <exp> 
 
<exp>                   ::= <fator> <comp> <fator> 
 
<fator>                 ::= <id> 
                         |  NUMBER 
                         |  boolean 
 
<comp>                  ::= EQ 
                         |  NE 
                         |  LT 
                         |  GT 
                         |  LE 
                         |  GE 
 
<decl_action>           ::= ACTION <action_body> END_ACTION 
                         |  ACTION <id> <action_body> END_ACTION 
 
<action_body>           ::= <action_elements> <action_body> 
                         |  <action_elements> 
 
<action_elements>       ::= <instigation> 
                         |  <method_use> 
                         |  <exp> SEMICOLON 
 
<instigation>           ::= INSTIGATION <method_use> 
                         |  INSTIGATION <id> <method_use> 
 
<method_use>            ::= <id> LP RP SEMICOLON 

Código 1: BNF da linguagem de programação LingPON Preliminar 
Fonte: Adaptado de Ronszcka et al., 2013 


